Street Flow (Shallow or Sheet Flow)

Revised 4/28/18 by ENS

This program calculates the depth of flow in a composite gutter section using Manning's equation modified for shallow or sheet flow using the HEC 22 (August 2013) equation that describes Q in terms of Sx, S, and T and not the equivalent equation that described Q in terms of Z, S, and d. Program assumes: (1) the curb face is vertical, (2) friction on the curb face is ignored, and (3) flow is contained in the street and gutter section, even if the water depth is above the height of the curb. There are two programs – one program that uses the same N value for both the gutter and street pavement (StreetFlow1N), and a second program that allows for different N values for the gutter and street pavement (StreetFlow2N).



Contents

A. Formula for shallow flow in a triangular channel given the top width T

The following formulas are from Urban Drainage Design Manual, Hydraulic Engineering Circular No. 22, 3rd ed., August 2013, Federal Highway Administration, page 4-9

$$Q = \frac{Ku}{n} * {S_x}^{1.67} * {S_L}^{0.5} * T^{2.67}$$

where Ku = 0.56 for English units, and Ku = 0.375 for metric units

$$T = \left[\frac{Qn}{k * {S_x}^{1.67} * {S_L}^{0.5}}\right]^{0.375}$$

$$d = T * S_x$$

Converting the decimal exponents back to their original fractions we get.

$$Q = \frac{Ku}{n} * {S_x}^{5/3} * {S_L}^{1/2} * T^{8/3}$$

Where Ku for metric units = 0.375, and Ku for English units = 0.56

From Stormwater Collection Systems Design Handbook, L. W. Mays, 2001, McGraw-Hill, page 5.9

$$S_g = S_p + \frac{gutter\ depression}{W_g}$$, or

Gutter depression $Gdepr = W_g * (S_g - S_p)$

B. Given depth Dg Find Q and T for a composite street section using similar triangles method

Given: Street longitudinal slope S = 0.035
Pavement cross slope Sp = 0.02
Pavement Manning Np = 0.016
Gutter width Wg = 2.5 ft
Gutter cross slope Sg = 0.05
Gutter Manning Ng = 0.014

Case 1. Flow depth is less than the height of the gutter (Dg<Hg), meaning flow is contained in the gutter.

Given flow depth Dg is 0.1 ft, find the flowrate Q and top width T

top width $Tg = \frac{Dg}{Sg}$

$$= \frac{0.1}{0.05} = 2\ ft$$

Flowrate $Q = \frac{Ku}{N_g} * S_g{}^{5/3} * S^{1/2} * T_g{}^{8/3}$

$$= \frac{0.56}{0.014} * 0.05^{\frac{5}{3}} * 0.035^{\frac{1}{2}} * 2^{\frac{8}{3}} = 0.3224\ cfs$$

Area $A = \frac{1}{2} * D_g * T_g$

$$= \frac{1}{2} * 0.1 * 2 = 0.1\ ft^2$$

Velocity $V = \frac{Q}{A}$

$$= \frac{0.3224}{0.1} = 3.224\ ft/s$$

Gutter depression $Gdepr = W_g * (S_g - S_p)$

$$= 2.5 * (0.05 - 0.02) = 0.075\ ft$$

Case 2. Flow depth greater or equal to the height of the gutter (Dg>=Hg).

Given flow depth Dg is 0.285 ft, find the flowrate Q and top width T

The gutter flowrate Qg is determined using similar triangles. The similar triangle method is discussed in the Urban Storm Drainage Criteria Manual, Volume 1, Urban Drainage and Flood Control District, Colorado, Jan 2016, page 7-10. As used in this program, the water flow is assumed to be contained in the curb section, even if the water elevation is higher than the top of curb.

Determine flowrate Q_{g1}

$$T1 = \frac{Dg}{S_g}$$

$$= \frac{0.285}{0.05} = 5.7\ ft$$

$$Q_{g1} = \frac{Ku}{N_g} * Sg^{5/3} * S^{1/2} * T1^{8/3}$$

$$= \frac{0.56}{0.014} * 0.05^{\frac{5}{3}} * 0.035^{\frac{1}{2}} * 5.7^{\frac{8}{3}} = 5.2647 cfs$$

Determine flowrate Q_{g1}

$$T2 = T1 - W_g$$

$$= 5.7 - 2.5 = 3.2\ ft$$

$$Q2 = \frac{Ku}{N_g} * Sg^{5/3} * S^{1/2} * T2^{8/3}$$

$$= \frac{0.56}{0.014} * 0.05^{\frac{5}{3}} * 0.035^{\frac{1}{2}} * 3.2^{\frac{8}{3}} = 1.1292\ cfs$$

Gutter flowrate $Q_g = Q_{g1} - Q_{g2}$

$$= 5.2647 - 1.1292 = 4.1355\ cfs$$

Determine pavement flowrate Qp

Gutter Height $H_g = W_g * S_g$

$$= 2.5 * 0.05 = 0.125\ ft$$

Pavement depth $D_p = D_g - H_g$

$$= 0.285 - 0.125 = 0.16\ ft$$

Pavement top width $Tp = \frac{Dp}{Sp}$

$$= \frac{0.16}{0.02} = 8\ ft$$

Pavement flowrate $Q_p = \frac{Ku}{N_P} * Sp^{5/3} * S^{1/2} * Tp^{8/3}$

$$= \frac{0.56}{0.016} * 0.02^{\frac{5}{3}} * 0.035^{\frac{1}{2}} * 8^{\frac{8}{3}} = 2.4701\ cfs$$

total flowrate $Q = Q_g + Q_p$

$$= 4.1355 + 2.4701 = 6.6056\ cfs$$

Top width $T = W_g + T_p$

$$= 2.5 + 8 = 10.5\ ft$$

Gutter area $A_g = \left(\frac{D_g + D_p}{2}\right) * W_g$

$$= \left(\frac{0.285 + 0.16}{2}\right) * 2.5 = 0.5562\ sf$$

Pavement area $A_p = \frac{1}{2} * D_p * T_p$

$$= \frac{1}{2} * 0.16 * 8 = 0.64\ sf$$

Total Area $A = A_g + A_p$

$$= 0.5562 + 0.64 = 1.1962\ sf$$

Velocity $V = \frac{Q}{A}$

$$= \frac{6.6056}{1.1962} = 5.5221\ fps$$

Gutter depression $Gdepr = W_g * (S_g - S_p)$

$$= 2.5 * (0.05 - 0.02) = 0.075\ ft$$

C. Spot Comparison with FHWA Hydraulic Toolbox Gutter Analysis program

The program was compared with the New Curb and Gutter Analysis Calculator in the Hydraulic Toolbox 4.3 program by the Federal Highways Admin., build date Oct 12, 2017. In this comparison, the Manning n value for the gutter and pavement section are the same.

Hydraulic Toolbox 4.3, composite gutter

Longitudinal Slope of Road S	Cross slope of Pavement Sp	Cross slope of gutter Sg	N	Gutter width Wg	Design Flow Q	Calc Width of Spread T	Calc Gut. Depr.	Calc Area of Flow A	Calc E0	Calc Depth at Curb Dg
0.01	0.017	0.09	0.016	2	3	10.683	1.752_in	1.116	0.583	3.931_in
							0.146_ft			0.3275_ft
0.01	0.017	0.09	0.015	2	6.7449	14.941	1.752_in	2.044	0.426	4.800_in
							0.146_ft			0.400_ft
0.008	0.02	.0909	0.017	1.5	5	13.556	1.276_in	1.917	0.337	4.530_in
							0.106_ft			0.378_ft
0.005	.02	.091	.016	2	0.02	0.734	1.704_ft	0.147	1	1.880_in
							0.142_ft			0.1566_ft

Hydraulic Toolbox 4.3, triangular street section, cross slope of pavement equal to gutter cross slope. Defined cross slope of gutter is unchecked.

Longitudinal Slope of Road S	Cross slope of Pavement Sp	Cross slope of gutter Sg	N	Gutter width Wg	Design Flow Q	Calc Width of Spread T	Calc Gut. Depr.	Calc Area of Flow A	Calc E0	Calc Depth at Curb Dg
0.005	0.091	NA	.016	2	0.02	0.734	0	0.025	1	0.802_in
										0.0668_ft

This program using Ku = 0.56

Street longitudinal slope S	Pavement cross slope Sp	Gutter cross slope Sg	N	Gutter width Wg	Flowrate Q	Calc top width T	Calc Gut. Depr.	Calc area A	Calc E0	Calc gutter depth Dg
0.01	0.017	0.09	0.016	2	3	10.683	0.146	1.116		0.328
0.01	0.017	0.09	0.015	2	6.7449	14.941	0.146	2.044		0.400
0.008	0.02	.0909	0.017	1.5	5	13.556	0.106	1.917		0.377

0.005	.02	.091	.016	2	0.02	0.734	0.142	0.025		0.067

The program results closely agrees with the FHWA Hydraulic Toolbox program, provided the flow depths exceed the gutter height.

For a composite gutter section, the FHWA Hydraulic Toolbox program does not correctly calculate flow depths and areas for depths less than the gutter height. For these flow depths, the "Defined Cross-slope of Gutter" needs to be unchecked, and the problem needs to be computed as a triangular street section whose pavement cross slope is equal to the gutter cross slope. When this is done, the FHWA Hydraulic Toolbox program results closely agrees the this program.

D. Metric Units - Given depth Dg Find Q and T for a composite street section using similar triangles method

Given: Street longitudinal slope S = 0.01
Pavement cross slope Sp = 0.017
Pavement Manning Np = 0.016
Gutter width Wg = 0.6 m
Gutter cross slope Sg = 0.09
Gutter Manning Ng = 0.014

Case 1. Flow depth is less than the height of the gutter (Dg<Hg), meaning flow is contained in the gutter.

Given flow depth Dg is 0.03 m, find the flowrate Q and top width T

top width $Tg = \frac{Dg}{Sg}$

$$= \frac{0.03}{0.09} = 0.3333\ m$$

Flowrate $Q = \frac{Ku}{n} * S_g{}^{5/3} * S^{1/2} * T_g{}^{8/3}$

$$= \frac{0.375}{0.014} * 0.09^{\frac{5}{3}} * 0.01^{\frac{1}{2}} * 0.3333^{\frac{8}{3}} = 0.002585\ m^3/s$$

Area $A = \frac{1}{2} * D_g * T_g$

$$= \frac{1}{2} * 0.03 * 0.3333 = 0.004999$$

Velocity $V = \frac{Q}{A}$

$$= \frac{0.002585}{0.004999} = 0.5171\ m/s$$

Gutter depression $Gdepr = W_g * (S_g - S_p)$

$$= 0.6 * (0.09 - 0.017) = 0.0438\ m$$

Case 2. Flow depth greater or equal to the height of the gutter (Dg>=Hg).

Given flow depth Dg is 0.13 m, find the flowrate Q and top width T

Determine flowrate Q_{g1}

$$T1 = \frac{Dg}{S_g}$$

$$= \frac{0.13}{0.09} = 1.4444\ m$$

$$Q_{g1} = \frac{Ku}{N_g} * Sg^{5/3} * S^{1/2} * T1^{8/3}$$

$$= \frac{0.375}{0.014} * 0.09^{\frac{5}{3}} * 0.01^{\frac{1}{2}} * 1.4444^{\frac{8}{3}} = 0.1291\ m^3/s$$

Determine flowrate Q_{g1}

$$T2 = T1 - W_g$$

$$= 1.4444 - 0.6 = 0.8444\ m$$

$$Q2 = \frac{Ku}{N_g} * Sg^{5/3} * S^{1/2} * T2^{8/3}$$

$$= \frac{0.375}{0.014} * 0.09^{\frac{5}{3}} * 0.01^{\frac{1}{2}} * 0.8444^{\frac{8}{3}} = 0.03084\ m^3/s$$

Gutter flowrate $Q_g = Q_{g1} - Q_{g2}$

$$= 0.1291 - 0.03084 = 0.09826\ m^3/s$$

Determine pavement flowrate Qp

Gutter Height $H_g = W_g * S_g$

$$= 0.6 * 0.09 = 0.054\ m$$

Pavement depth $D_p = D_g - H_g$

$$= 0.13 - 0.054 = 0.076\ m$$

Pavement top width $Tp = \frac{Dp}{Sp}$

$$= \frac{0.076}{0.017} = 4.4706\ m$$

Pavement flowrate $Q_p = \frac{Ku}{N_p} * Sp^{5/3} * S^{1/2} * Tp^{8/3}$

$$= \frac{0.375}{0.016} * 0.017^{\frac{5}{3}} * 0.01^{\frac{1}{2}} * 4.4706^{\frac{8}{3}} = 0.1429\ m^3/s$$

total flowrate $Q = Q_g + Q_p$

$$= 0.09826 + 0.1429 = 0.2412\ m^3/s$$

Top width $T = W_g + T_p$

$$= 0.6 + 4.4706 = 5.0706\ m$$

Gutter area $A_g = \left(\frac{D_g + D_p}{2}\right) * W_g$

$$= \left(\frac{0.13 + 0.076}{2}\right) * 0.6 = 0.0618\ m\text{^}2$$

Pavement area $A_p = \frac{1}{2} * D_p * T_p$

$$= \frac{1}{2} * 0.076 * 4.4706 = 0.1699\ m\text{^}2$$

Total Area $A = A_g + A_p$

$$= 0.0618 + 0.1699 = 0.2317\ m\text{^}2$$

Velocity $V = \frac{Q}{A}$

$$= \frac{0.2412}{0.2317} = 1.0414\ m/s$$

Gutter depression $Gdepr = W_g * (S_g - S_p)$

$$= 0.6 * (0.09 - 0.017) = 0.0438\ m$$

E. Program Listing for StreetFlow1N

```
// Street Flow 1N (Shallow or Sheet Flow)
// 4/28/18 by ENS

//Declare subroutines and functions
subQ();

//Variable declarations
EXPORT Dg,Wg,Sg,Sp,Qg,Qp,SolveNum,UnitsSF;
LOCAL Tg,Dp,Tp,Hg,Gdepr,Hc;
LOCAL Ku,Q1,T1,Q2,T2,Dgx,Qx,Tx;
LOCAL Ax,Ag,Ap,Qg1,Qg2;
```

```
LOCAL Da,fDa,Db,fDb,Dc,fDc;  //bisection method variables
LOCAL CFScms,SFsm,FTm,FPSmps,SolvedQ,SolvedT,SolvedDg;  //unit labels used when
printing

EXPORT StreetFlow1N()
BEGIN

INPUT(
  {
    {UnitsSF,{"English Units","Metric Units"},{1,35,0}},
    {S,[0],{30,20,1}},
    {Sp,[0],{30,20,2}},
    {N,[0],{30,20,3}},
    {Wg,[0],{80,20,1}},
    {Sg,[0],{80,20,2}},
    {SolveNum,{"Enter flowrate Q and solve T,Dg","Enter width T and solve Q,Dg",
"Enter depth Dg and solve Q,T"},{1,70,4}},
    {Q,[0],{30,20,5}},
    {T,[0],{80,20,5}},
    {Dg,[0],{30,20,6}}
  },
  "Street Flow 1N (Shallow or Sheet Flow)",
  {
    " ",
    "street slope ",
    "pvmt x-slope ",
    "Manning N ",
    "gutter width ",
    "gut. x-slope ",
    " ",
    "flowrate Q ", "top width T ",
    "depth Dg "
  },
  {
    "Select English or Metric Units",
    "street longitudinal slope S (ft/ft, m/m)",
    "pavement cross slope Sp (ft/ft, m/m)",
    "Manning N value (N)",
    "gutter width Wg (ft, m)",
    "gutter cross slope Sg (ft/ft, m/m)",
    "Select which 2 variables to solve for",
    "flowrate Q (cfs, m^3/s)",
    "top width T (ft, m)",
    "depth Dg (ft, m)"
   });
//  Input format that was used
//  {variable,[0],{left%,length%,row number}, where [0] is for real numbers, left% is
for the right edge of the input box
//  {variable,{"Text1","Text2"},{left%,length%,row number}, format for a dropbox aka
choose box

//English and Metric Unit definitions
CASE
  //Case 1 English Units
  IF UnitsSF=1 THEN
```

```
    Ku:=0.56;
    //Unit labels for printing the results
    CFScms:=" cfs";
    FTm:=" ft";
    FPSmps:=" ft/s";
    SFsm:=" ft^2";
    SolvedQ:="";
    SolvedT:="";
    SolvedDg:="";
  END;
  //Case 2 Metric Units
  IF UnitsSF=2 THEN
    Ku:=0.375;
    //Unit labels for printing the results
    CFScms:=" m^3/s";
    FTm:=" m";
    FPSmps:=" m/s";
    SFsm:=" m^2";
    SolvedQ:="";
    SolvedT:="";
    SolvedDg:="";
  END;
END;  // end case

//gutter height Hg
Hg:=Wg*Sg;
//gutter depresion Gdepr
Gdepr:=Wg*(Sg-Sp);

CASE
  //Case 1 Given flowrate Q, find depth Dg,T, using bisection method
  IF SolveNum=1 THEN
    //define initial bisection endpoints: depth Da and depth Db
    Da:=0.001;        //endpoint a
    Db:=100;          //endpoint b
    //loop for bisection method
    FOR I FROM 1 TO 100 DO
      Dgx:=Da;
      subQ();         //given Dgx = endpoint a, find Qx
      fDa:=Qx-Q;
      Dgx:=Db;
      subQ();         //given Dgx = endpoint b, find Qx
      fDb:=Qx-Q;
      Dc:=(Da+Db)/2;  //midpoint c
      Dgx:=Dc;
      subQ();         //given Dgx = midpoint c, find Qx,Tx,Ax
      fDc:=Qx-Q;
      IF SIGN(fDc)=SIGN(fDa) THEN
        Da:=Dc;       //new endpoint a
      ELSE
        Db:=Dc;       //new endpoint b
      END;
      IF (ABS(Db-Da)/2)<0.0000001 THEN
        Dg:=Dc;       //save last depth
        T:=Tx;        //save top width for last depth
```

```
        A:=Ax;        //save area Ax for last depth
        BREAK(1);
      END;
    END; //for loop
    V:=Q/A;
    SolvedT:="** ";   //flag T variable when printing
    SolvedDg:="** ";  //flag Dg variable when printing
  END; //case 1

  //Case 2 Given T Find Q,Dg
  IF SolveNum=2 THEN
    //determine depth Dg
    IF T>Wg THEN
      //depth Dg is greater than the gutter height (composite gutter section)
      Tp:=T-Wg;
      Dp:=Tp*Sp;
      Dg:=Hg+Dp;
    ELSE
      //depth Dg is less than the gutter height (triangular gutter section)
      Dg:=T*Sg;
    END;
    Dgx:=Dg;
    subQ();
    Q:=Qx;            //given Dgx=Dg, find Qx,Tx,Ax
    T:=Tx;
    A:=Ax;
    V:=Q/A;
    SolvedQ:="** ";   //flag Q variable when printing
    SolvedDg:="** ";  //flag Dg variable when printing
  END; //case 2

  //Case 3 Given Dg Find Q,T
  IF SolveNum=3 THEN
    Dgx:=Dg;
    subQ();           //given Dgx=Dg, find Qx,Tx,Ax
    Q:=Qx;
    T:=Tx;
    A:=Ax;
    V:=Q/A;
    SolvedQ:="** ";   //flag Q variable when printing
    SolvedT:="** ";   //flag T variable when printing
  END; //case 3

END;  //case

//print output
PRINT();
PRINT("Street Flow 1N (Shallow or Sheet Flow)");
PRINT(" ");
PRINT(SolvedQ+"flowrate Q = "+ROUND(Q,3)+CFScms);
PRINT(SolvedT+"top width T = "+ROUND(T,3)+FTm);
PRINT(SolvedDg+"gutter depth Dg = "+ROUND(Dg,3)+FTm);
PRINT("area A = "+ROUND(A,4)+SFsm);
PRINT("velocity V = "+ROUND(V,3)+FPSmps);
PRINT("street longitudinal slope S = "+ROUND(S,6));
```

```
PRINT("pavement cross slope Sp = "+ROUND(Sp,6));
PRINT("Manning N = "+N);
PRINT("gutter width Wg = "+ROUND(Wg,3)+FTm);
PRINT("gutter cross slope Sg = "+ROUND(Sg,6));
//PRINT("gutter height Hg = "+ROUND(Hg,3)+FTm);
PRINT("gutter flowrate Qg = "+ROUND(Qg,3)+CFScms);

END;  //main program

//subroutines

//Calculate the flowrate Qx, top width Tx, and area Ax, given depth Dgx for a
composite gutter section
//This subroutine is used like a function, except that Dgx has to be manually inputted
into the
//subroutine and Qx, Tx, and Ax are manually retrieved from the subroutine.
subQ()
BEGIN
  //Flow is contained in gutter
  //depth Dgx is less than the gutter height Hg
  IF Dgx<Hg THEN
    Tg:=Dgx/Sg;
    Qx:=(Ku/N)*Sg^(5/3)*S^(1/2)*Tg^(8/3);
    Tx:=Tg;
    Ax:=0.5*Dgx*Tg;
  END;
  //Flow is contained in curb
  //depth Dgx is greater or equal to the gutter height Hg
  IF Dgx>=Hg THEN
    T1:=Dgx/Sg;
    Qg1:=(Ku/N)*Sg^(5/3)*S^(1/2)*T1^(8/3);
    T2:=T1-Wg;
    Qg2:=(Ku/N)*Sg^(5/3)*S^(1/2)*T2^(8/3);
    Qg:=Qg1-Qg2;
    Dp:=Dgx-Hg;
    Tp:=Dp/Sp;
    Qp:=(Ku/N)*Sp^(5/3)*S^(1/2)*Tp^(8/3);
    Qx:=Qg+Qp;
    Tx:=Wg+Tp;
    Ag:=Wg*(Dgx+Dp)/2;
    Ap:=0.5*Dp*Tp;
    Ax:=Ag+Ap;
  END;
END; //subroutine
```

F. Program Listing for StreetFlow2N

```
// Street Flow 2N (Shallow or Sheet Flow)
// 4/28/18 by ENS

//Declare subroutines and functions
subQ();
```

```
//Variable declarations
EXPORT Dg,Wg,Sg,Sp,Qg,Qp,Ng,Np;
EXPORT SolveNum,UnitsSF;
LOCAL Tg,Dp,Tp,Hg,Gdepr,Hc;
LOCAL Ku,Q1,T1,Q2,T2,Dgx,Qx,Tx;
LOCAL Ax,Ag,Ap,Qg1,Qg2;
LOCAL Da,fDa,Db,fDb,Dc,fDc;  //bisection method variables
LOCAL CFScms,SFsm,FTm,FPSmps,SolvedQ,SolvedT,SolvedDg;  //unit labels used when
printing

EXPORT StreetFlow2N()
BEGIN

INPUT(
  {
    {UnitsSF,{"English Units","Metric Units"},{1,35,0}},
    {S,[0],{30,20,1}},
    {Sp,[0],{30,20,2}},
    {Np,[0],{30,20,3}},
    {Wg,[0],{80,20,1}},
    {Sg,[0],{80,20,2}},
    {Ng,[0],{80,20,3}},
    {SolveNum,{"Enter flowrate Q and solve T,Dg","Enter width T and solve Q,Dg",
"Enter depth Dg and solve Q,T"},{1,70,4}},
    {Q,[0],{30,20,5}},
    {T,[0],{80,20,5}},
    {Dg,[0],{30,20,6}}
  },
  "Street Flow 2N (Shallow or Sheet Flow)",
  {
    " ",
    "street slope ",
    "pvmt x-slope ",
    "pavement N ",
    "gutter width ",
    "gut. x-slope ",
    "gutter N ",
    " ",
    "flowrate Q ", "top width T ",
    "depth Dg "
  },
  {
    "Select English or Metric Units",
    "street longitudinal slope S (ft/ft, m/m)",
    "pavement cross slope Sp (ft/ft, m/m)",
    "pavement Manning N value (Np)",
    "gutter width Wg (ft, m)",
    "gutter cross slope Sg (ft/ft, m/m)",
    "gutter Manning N value (Ng)",
    "Select which 2 variables to solve for",
    "flowrate Q (cfs, m^3/s)",
    "top width T (ft, m)",
    "depth Dg (ft, m)"
   });
//  Input format that was used
```

```
//  {variable,[0],{left%,length%,row number}, where [0] is for real numbers, left% is
for the right edge of the input box
//  {variable,{"Text1","Text2"},{left%,length%,row number}, format for a dropbox aka
choose box

//English and Metric Unit definitions
CASE
  //Case 1 English Units
  IF UnitsSF=1 THEN
    Ku:=0.56;
    //Unit labels for printing the results
    CFScms:=" cfs";
    FTm:=" ft";
    FPSmps:=" ft/s";
    SFsm:=" ft^2";
    SolvedQ:="";
    SolvedT:="";
    SolvedDg:="";
  END;
  //Case 2 Metric Units
  IF UnitsSF=2 THEN
    Ku:=0.375;
    //Unit labels for printing the results
    CFScms:=" m^3/s";
    FTm:=" m";
    FPSmps:=" m/s";
    SFsm:=" m^2";
    SolvedQ:="";
    SolvedT:="";
    SolvedDg:="";
  END;
END;  // end case

//gutter height Hg
Hg:=Wg*Sg;
//gutter depresion Gdepr
Gdepr:=Wg*(Sg-Sp);

CASE
  //Case 1 Given flowrate Q, find depth Dg,T, using bisection method
  IF SolveNum=1 THEN
    //define initial bisection endpoints: depth Da and depth Db
    Da:=0.001;        //endpoint a
    Db:=100;          //endpoint b
    //loop for bisection method
    FOR I FROM 1 TO 100 DO
      Dgx:=Da;
      subQ();         //given Dgx = endpoint a, find Qx
      fDa:=Qx-Q;
      Dgx:=Db;
      subQ();         //given Dgx = endpoint b, find Qx
      fDb:=Qx-Q;
      Dc:=(Da+Db)/2;  //midpoint c
      Dgx:=Dc;
      subQ();         //given Dgx = midpoint c, find Qx,Tx,Ax
```

```
      fDc:=Qx-Q;
      IF SIGN(fDc)=SIGN(fDa) THEN
        Da:=Dc;       //new endpoint a
      ELSE
        Db:=Dc;       //new endpoint b
      END;
      IF (ABS(Db-Da)/2)<0.0000001 THEN
        Dg:=Dc;       //save last depth
        T:=Tx;        //save top width for last depth
        A:=Ax;        //save area Ax for last depth
        BREAK(1);
      END;
    END; //for loop
    V:=Q/A;
    SolvedT:="** ";   //flag T variable when printing
    SolvedDg:="** ";  //flag Dg variable when printing
  END; //case 1

  //Case 2 Given T Find Q,Dg
  IF SolveNum=2 THEN
    //determine depth Dg
    IF T>Wg THEN
      //depth Dg is greater than the gutter height (composite gutter section)
      Tp:=T-Wg;
      Dp:=Tp*Sp;
      Dg:=Hg+Dp;
    ELSE
      //depth Dg is less than the gutter height (triangular gutter section)
      Dg:=T*Sg;
    END;
    Dgx:=Dg;
    subQ();
    Q:=Qx;            //given Dgx=Dg, find Qx,Tx,Ax
    T:=Tx;
    A:=Ax;
    V:=Q/A;
    SolvedQ:="** ";   //flag Q variable when printing
    SolvedDg:="** ";  //flag Dg variable when printing
  END; //case 2

  //Case 3 Given Dg Find Q,T
  IF SolveNum=3 THEN
    Dgx:=Dg;
    subQ();           //given Dgx=Dg, find Qx,Tx,Ax
    Q:=Qx;
    T:=Tx;
    A:=Ax;
    V:=Q/A;
    SolvedQ:="** ";   //flag Q variable when printing
    SolvedT:="** ";   //flag T variable when printing
  END; //case 3

END;  //case

//print output
```

```
PRINT();
PRINT("Street Flow 2N (Shallow or Sheet Flow)");
PRINT(" ");
PRINT(SolvedQ+"flowrate Q = "+ROUND(Q,3)+CFScms);
PRINT(SolvedT+"top width T = "+ROUND(T,3)+FTm);
PRINT(SolvedDg+"gutter depth Dg = "+ROUND(Dg,3)+FTm);
PRINT("area A = "+ROUND(A,4)+SFsm);
PRINT("velocity V = "+ROUND(V,3)+FPSmps);
PRINT("street longitudinal slope S = "+ROUND(S,6));
PRINT("pavement cross slope Sp = "+ROUND(Sp,6));
PRINT("pavement Manning N (Np) = "+Np);
PRINT("gutter width Wg = "+ROUND(Wg,3)+FTm);
PRINT("gutter cross slope Sg = "+ROUND(Sg,6));
PRINT("gutter Manning N (Ng) = "+Ng);
//PRINT("gutter depression Gdepr = "+ROUND(Gdepr,3)+FTm);
//PRINT("gutter height Hg = "+ROUND(Hg,2)+FTm);

END;  //main program

//subroutines

//Calculate the flowrate Qx, top width Tx, and area Ax, given depth Dgx for a
composite gutter section
//This subroutine is used like a function, except that Dgx has to be manually inputted
into the
//subroutine and Qx, Tx, and Ax are manually retrieved from the subroutine.
subQ()
BEGIN
  //Flow is contained in gutter
  //depth Dgx is less than the gutter height Hg
  IF Dgx<Hg THEN
    Tg:=Dgx/Sg;
    Qx:=(Ku/Ng)*Sg^(5/3)*S^(1/2)*Tg^(8/3);
    Tx:=Tg;
    Ax:=0.5*Dgx*Tg;
  END;
  //Flow is contained in curb
  //depth Dgx is greater or equal to the gutter height Hg
  IF Dgx>=Hg THEN
    T1:=Dgx/Sg;
    Qg1:=(Ku/Ng)*Sg^(5/3)*S^(1/2)*T1^(8/3);
    T2:=T1-Wg;
    Qg2:=(Ku/Ng)*Sg^(5/3)*S^(1/2)*T2^(8/3);
    Qg:=Qg1-Qg2;
    Dp:=Dgx-Hg;
    Tp:=Dp/Sp;
    Qp:=(Ku/Np)*Sp^(5/3)*S^(1/2)*Tp^(8/3);
    Qx:=Qg+Qp;
    Tx:=Wg+Tp;
    Ag:=Wg*(Dgx+Dp)/2;
    Ap:=0.5*Dp*Tp;
    Ax:=Ag+Ap;
  END;
END; //subroutine
```